Estimating Social Animal Models in Stan

Jordan S. Martin & Adrian V. Jaeggi


2021 - 01 - 06

Contents

Introduction			5
1	Using Stan in R		
	1.1	Why Stan?	7
	1.2	Getting Started	8
	1.3	Bayesian inference	8
	1.4	Basic coding tutorial	9
	1.5	Animal models	21
2	SAM coding tutorial		
	2.1	Coding the model	31
	2.2	Quantifying assortment	31
	2.3	Selection differentials	31
	2.4	The response to selection	31
3	Ext	ending SAMs	33

4 CONTENTS

Introduction



This guidebook provides a comprehensive overview of how to estimate social animal models (SAMs) with the Stan statistical programming language (Carpenter et al., 2017) in R (R Core Team, 2020). A detailed theoretical treatment of SAMs and their empirical motivation can be found in the forthcoming paper (Martin and Jaeggi, 2021) "Social animal models for quantifying plasticity, assortment, and selection on interacting phenotypes" in the *Journal of Evolutionary Biology*. This guide focuses on additional issues related to coding SAMs in Stan, as well as the extension of SAMs beyond the Gaussian response models presented in the main text. The overarching goal of the guidebook is to aid researchers in flexibly and appropriately applying SAMs to their own empirical datasets. Therefore, in addition to basic coding tutorials, the guide also includes a growing body of worked examples relevant to specific challenges such as modeling spatial autocorrelation or hierarchical phenotypes.

Please contact Jordan Scott Martin to suggest any additional examples of interest to your own work. The guidebook is a work in progress and will be continuously updated over time.

6 CONTENTS

Chapter 1

Using Stan in R

1.1 Why Stan?

SAMs cannot be straightforwardly implemented with currently available software for quantitative genetic analysis, such as the frequentist ASREML program (Butler et al., 2018) or the Bayesian open-source R package MCMCglmm (Hadfield, 2010). The classical animal models estimated by these programs can be used to describe reaction norms defined over non-social environments, with reaction norm slopes estimated on directly measured environmental gradients. However, social environments defined by partner phenotypes present novel challenges for animal models, such as accounting for temporal feedback between social partners' phenotypes, differentiating the effects of assortment and social plasticity between partners, and avoiding bias due to correlated residual effects on measurements taken within and among social interactions (Martin and Jaeggi, 2021). SAMs address these challenges by estimating plasticity, assortment, and selection directly on the latent social reaction norms (SRNs) governing repeatable individual variation. A highly flexible modeling framework is required to estimate these latent (i.e. indirectly measured) interactions with raw empirical data, as well as to use them for predicting social evolutionary change.

Stan (Carpenter et al., 2017) is an open-source programming language for estimating probabilistic models of arbitrary complexity, which can interface with multiple statistical environments such as R (R Core Team, 2020). Stan also facilitates fully Bayesian inference using state-of-the-art Markov Chain Monte Carlo (MCMC) sampling techniques. In particular, the No U-Turn Sampler (NUTS) implimented in Stan has been found to perform particularly well for quantitative genetic analysis (Nishio and Arakawa, 2019). Stan is thus an ideal platform for flexibly estimating SAMs in any empirical system, as is further discussed in the main text (Martin and Jaeggi, 2021). Using Bayesian posteriors rather than point estimates will also promote more robust biological inferences

with SAMs, as statistical uncertainty can be easily carried forward across multiple stages of analysis (Stinchcombe et al., 2014). This provides a crucial means of quantifying uncertainty in the predicted direction and magnitude of social evolution.

1.2 Getting Started

Stan interfaces with R through the RStan package (Carpenter et al., 2017), providing an efficient means of integrating SAMs into pre-existing data analysis pipelines. However, you will first need to install Stan on your computer and ensure that it is appropriately configured with your C++ toolchain. This can be accomplished by following the instructions for your operating system on the RStan Getting Started page. Once you are able to effectively use RStan, you can begin creating the .stan files necessary for estimating SAMs. These files can be composed using RStudio or any text editor, as explained below. Once an appropriate .stan file is prepared, it can be compiled with R for the C++ toolchain using the stan_model() function and subsequently estimated with an appropriate list of empirical data using the sampling() function. The resulting posteriors of a model can then be accessed with the extract() function and manipulated for any further quantities or analyses of interest.

```
library(rstan)
#compiles the model in C++ for MCMC estimation
model = stan_model("model.stan")

#estimates and saves the posterior MCMC samples of the model (estimated with default s
results = sampling(object = SAM_file, data = data)

#extracts posterior estimates
MCMCsamples = extract(results)
```

1.3 Bayesian inference

A detailed overview of the benefits of Bayesian inference is beyond the scope of this guidebook, as attention is placed on coding and computational concerns rather than interpretation. We encourage researchers unfamiliar with fully Bayesian inference to see McElreath (2020) for further discussion. Lemoine (2019) also demonstrates why weakly regularizing (or "weakly informative") priors are often preferable to the flat or diffuse priors more commonly used in evolutionary ecology. In general, we encourage researchers to utilize the benefits of fully Bayesian inference while working in Stan, rather than attempting

to mimic classical inference and null-hypothesis testing approaches. Gelman et al. (2020) provide a very useful general discussion of Bayesian workflow from initial estimation to model comparison and selection. A basic understanding of MCMC and prior and posterior distributions is necessary to fully understand model estimation in Stan. MCMC provides a means of approximating any continous probability distribution, with a finite set of samples taken in proportion to the underlying target probability density. As a consequence, Stan models return objects with many MCMC samples for each model parameter, rather than single point estimates. These samples can then be summarized to approximate the shape of the truly continous target posterior distribution, as is shown throughout the coding tutorials.

1.4 Basic coding tutorial

Stan uses its own language for writing probabilistic models, including a variety of built-in functions designed to aid in efficient computation. The biggest conceptual hurdle for new users of Stan is likely to be the absence of an intuitive R-like syntax for specifying model formulas, such as formulas like y ~ x + (1|z) that can be used to quickly specify complex generalized linear mixedeffects models. These formulas facilitate highly efficient statistical modeling, but do so at the cost of limiting users' ability to specify atypical model structures. Instead, Stan provides the benefit of nearly unlimited flexibility in model specification, with the added cost of a steeper learning curve. In particular, models must be formally specified with mathematically appropriate likelihood functions, rather than this process being handled on the back-end through textual inputs from the user such as family= poisson(link = "log"). This may at first seem like a cumbersome task, but it provides a degree of independence and creativity for data analysis that is otherwise unavailable. It is this autonomy that makes it possible to unbiasedly estimate SAMs in Stan, which to the best of our knowledge cannot be accomplished with any other mainstream statistical software. Nonetheless, it is important to recognize that some practice and trial-and-error will be required to gain competency and comfortability with Stan. We therefore encourage those interested in SAMs to review the Stan Reference Manual, as well the extensive collection of Stan Case Studies, which will provide a more robust foundation for estimating any model of interest in Stan.

Here we review some basics of Stan that will be necessary for following the coding tutorials in the rest of the guidebook. To make this introduction more concrete, we simulate a simple data structure appropriately described by a Gaussian random regression model, with 50 subjects and 2 repeated measures per subject across an environmental gradient x. Formally, the model for observation i of individual j is given by

$$y_{ij} = \mu_j + \beta_j x_{ij} + \epsilon_i$$

$$\mu_j = \mu_0 + \mu_{Pj}, \quad \beta_j = \beta_1 + \beta_{Pj}$$

$$\begin{bmatrix} \mu_{\mathbf{P}} \\ \beta_{\mathbf{P}} \end{bmatrix} \sim \text{MVNormal}(0, \mathbf{P}) : \mathbf{P} = \begin{bmatrix} \text{Var}(\mu_{\mathbf{P}}) & \text{Cov}(\mu_{\mathbf{P}}, \beta_{\mathbf{P}}) \\ \text{Cov}(\beta_{\mathbf{P}}, \mu_{\mathbf{P}}) & \text{Var}(\beta_{\mathbf{P}}) \end{bmatrix}$$

$$\epsilon \sim \text{Normal}(0, \mathbf{R}) : \mathbf{R} = [\text{Var}(\epsilon)]$$

where μ_0 and β_1 are fixed population-level intercepts and slopes respectively, with the vectors $\mu_{\mathbf{P}}$ and $\beta_{\mathbf{P}}$ containing individual-specific phenotypic deviations from the population values (i.e. random intercepts and slopes). The probability density function of this Gaussian variable can be equivalently written as

$$y_{ij} \sim \text{Normal}(\mu_j + \beta_j x_{ij}, \text{Var}(\epsilon))$$

It is often easier to specify model likelihoods and priors over standard deviations and correlation matrices in Stan, rather than the variances and covariances represented in the formal model. These parameters can always be derived from one another with simple transformations. For variances and standard deviations

$$\mathrm{SD}(\mu_{\mathbf{P}}) = \mathrm{Sqrt}(\mathrm{Var}(\mu_{\mathbf{P}})), \quad \mathrm{SD}(\beta_{\mathbf{P}}) = \mathrm{Sqrt}(\mathrm{Var}(\beta_{\mathbf{P}}))$$

Similarly, the covariance matrix \mathbf{P} can be derived by pre- and post-multiplying the correlation matrix $\mathbf{P_{Cor}}$ with diagonal matrices $\mathbf{P_{SD}}$ of these standard deviations

$$\mathbf{P} = \mathbf{P_{SD}} \mathbf{P_{Cor}} \mathbf{P_{SD}}$$

$$\mathbf{P_{SD}} = \begin{bmatrix} \mathrm{SD}(\mu_{\mathbf{P}}) & 0 \\ 0 & \mathrm{SD}(\beta_{\mathbf{P}}) \end{bmatrix}, \quad \mathbf{P_{Cor}} = \begin{bmatrix} 1 & \mathrm{Cor}(\mu_{\mathbf{P}}, \beta_{\mathbf{P}}) \\ \mathrm{Cor}(\beta_{\mathbf{P}}, \mu_{\mathbf{P}}) & 1 \end{bmatrix}$$

We can simulate a random dataset from this model in R, along with an index variable id that tracks which individual (I = 1 - 50) is being measured at each observation (N = 1-100).

```
library(mvtnorm)

N = 100 #total observations
I = 50 #total individuals

intercept = 1 #global intercept
beta1 = 0.3 #fixed effect regression coefficient
SD_intercept = 0.3 #standard deviation of random intercepts
SD slope = 0.3
```

```
SD_residual = 1
cor_RE = 0.3 #correlation of random intercepts and slopes
#individual-level index
id = rep(seq(1, I), each = N/I) #i.e. two observations per individual
#simulate fixed effect covariate
x = rnorm(100,0,1) #fixed effect covariate
#simulate random individual deviations
P_cor = matrix( c(1, cor_RE, cor_RE, 1), nrow = 2, ncol = 2 )
P_SD = matrix( c(SD_intercept, 0, 0, SD_slope), nrow = 2, ncol = 2 )
P_cov = P_SD %*% P_cor %*% P_SD
re_P = rmvnorm(I, mean = c(0,0), sigma = P_cov) #rows = I, cols = intercepts and slopes
#individual-level parameters
mu = intercept + re_P[,1]
beta = beta1 + re_P[,2]
#residual effects
epsilon = rnorm(100, 0, SD_residual )
#measured response (100 response values for 50 subjects)
y = mu[id] + beta[id]*x + epsilon
#combine into list for Stan
#other values are empirically unobserved and will be model parameters
stan_data = list(y = y, x = x, id = id, N = N, I = I)
```

We can now program a Stan model to infer the data-generating process with these empirical observations. For any .stan file composed with a text editor, the following programming blocks will be recognized and all model code inside each block will be processed sequentially.

```
functions {
}
data {
}
transformed data {
}
parameters {
}
transformed parameters {
}
model {
```

```
}
generated quantities {
}
```

Only the data, parameters, and model blocks are necessary for model estimation, while the other blocks provide optional declarations and statements. In most statistical software, empirical data are input with a single matrix or dataframe. Rather than inputting a single dataframe or matix to RStan, a list can be provided with data for each scalar (real or integer), vector, or matrix declared in the .stan file. The names of these data objects are declared along with their expected dimensions, which ensures that inappropriate data structures or likelihood functions will throw errors. For the simulated data, we first declare all the measured variables and indices relevant to model estimation. We use // rather than # for comments in Stan.

```
data {
  int<lower=0> N; //length of response vector/total observations; values less than 1 to
  int<lower=0> I; //number of individuals
  int<lower=0> id[N]; //N integer indices matching observations of y to the individual
  vector[N] x; //vector of covariate values for fixed effect
  vector[N] y; //vector of response values
}
```

This declarative approach requires that particular attention is given to the order of data input to the model, as values will need to be appropriately aligned and indexed throughout the model specification. However, it also provides additional benefits such as facilitating multi-response models with heterogeneous dimensions, as well as allowing for arbitrarily complex forms of social interaction to be specified in the model likelihood using appropriate indices of the relevant vectors or matrices.

We specify model parameters in accordance with the formal model used for the simulation, with standard deviations and correlation matrices replacing variances and covariance matrices.

```
parameters {
   //fixed effects
   real mu_0; //global intercept
   real beta_1; //fixed effect coefficient for covariate x

   //random effects
   corr_matrix[2] P_cor; //correlation matrix of random effects
   vector<lower=0>[2] sd_P; //standard deviations of random effects
   real<lower=0> sd_R; //standard deviation of residuals
```

```
matrix[I,2] re_P; //individual-level phenotypic deviations (random intercepts and slopes)
}
```

Note that rather than declaring the random effects as separate vectors, we instead declare a matrix for both individual intercept and slope values. It is necessary to specify <lower=0> so that the standard deviation parameters are lower bound at zero.

The other parameters in the formal model are simply combinations of these fundamental parameters. In particular, the individual-level intercepts μ_j and slopes β_j are determined by the sum of the population-level intercepts μ_0 and slopes β_1 with the respective random individual deviations $\mu_{\rm Pj}$ and $\beta_{\rm Pj}$, which are contained in the re_P matrix. Similarly, the covariance matrix P can be derived with the standard deviations sd_P and the correlation matrix 'P_cor as shown above. These parameters can be useful for increasing model clarity, as well for enhancing the efficiency of MCMC sampling as demonstrated further below. The transformed parameters block of a .stan file is intended for such purposes.

```
transformed parameters {
  vector[I] mu = mu_0 + col(re_P, 1); //individual-level intercepts
  vector[I] beta = beta_1 + col(re_P, 2); //individual-level intercepts
  cov_matrix[2] P = diag_matrix(sd_P) * P_cor * diag_matrix(sd_P); //cov of random effects
}
```

This code creates two new vectors mu and beta of length I containing the expected intercepts and slopes of each individual, as well as covariance matrix P derived from the standard deviations and correlation matrix. These quantities can now be used in the model block to more clearly express the likelihood function. Note that these new objects could also be declared in the model block prior to specifying the likelihood. However, any objects created in the model block are temporary and will not be saved along with the MCMC samples of objects declared in the parameters and transformed parameters blocks. This can be useful for creating pragmatic objects that enable more efficient coding but do not need to be directly interpreted.

Following the formal model above, we specify the response y_{ij} as a function of the linear predictor containing individual intercepts μ_j and slopes in response to the environmental covariate $\beta_j x_{ij}$, as well as stochastic effects with standard deviation $\mathrm{SD}(\epsilon) = \mathrm{SD}_{\mathrm{R}}$. The random effects are sampled from a zero-centered multivariate normal with covariance matrix \mathbf{P} .

```
model {
   //model likelihood
   y ~ normal(mu[id] + beta[id].*x, sd R); //index by id to match response vector length
```

```
for(i in 1:I)
re_P[i] ~ multi_normal([0,0], P );

//priors

//fixed effects
mu_0 ~ normal(0,1);
beta_1 ~ normal(0,1);

//random effects
P_cor ~ lkj_corr(2);
to_vector(sd_P) ~ cauchy(0,1);
sd_R ~ cauchy(0,1);
}
```

Model priors are set for all parameters declared in the original programming block, while transformed parameters do not receive priors. We use general purpose, weakly regularizing priors to reduce the risk of inferential bias and enhance model identification, which will be crucial for SAMs relying on interactions among many latent variables. Interested readers should see Lemoine (2019) and McElreath (2020) for further discussion on the choice of model priors, as well as the clear limitations of using highly diffuse, flat, and/or improper priors that are more commonly utilized. Finally, rather than post-processing the posterior SDs ourselves to derive variances, we can instead use the generated quantities block to calculate the variances during model estimation.

```
generated quantities {
  vector[2] V_P = sd_P .* sd_P;
  real V_R = sd_R * sd_R;
}
```

The posterior object returned from this model will now contain the random effects variances and covariance matrix, along with the SDs and correlation matrix. Each of these blocks can be saved together in a single .stan file with an empty line break left at the end of the file. This can be accomplished with a text editor or inside R.

```
data {
  int<lower=0> N; //length of response/total observations; values < 1 throw error
  int<lower=0> I; //number of individuals
  int<lower=0> id[N]; //matches observations of y to the individual identity

vector[N] x; //vector of covariate values for fixed effect
  vector[N] y; //vector of response values
```

```
}
parameters {
  //fixed effects
  real mu_0; //global intercept
  real beta 1; //fixed effect coefficient for covariate x
  //random effects
  corr_matrix[2] P_cor; //correlation matrix of random effects
  vector<lower=0>[2] sd_P; //standard deviations of random effects
  real<lower=0> sd R; //standard deviation of residuals
  matrix[I,2] re_P; //individual-level phenotypic deviations (random intercepts and slopes)
}
transformed parameters {
  vector[I] mu = mu_0 + col(re_P, 1); //individual-level intercepts
  vector[I] beta = beta_1 + col(re_P, 2); //individual-level intercepts
  cov_matrix[2] P = diag_matrix(sd_P) * P_cor * diag_matrix(sd_P); //cov of random effects
model {
  //model likelihood
  y ~ normal(mu[id] + beta[id].*x, sd_R); //index by id to match response vector length
  for(i in 1:I)
  re_P[i] ~ multi_normal([0,0], P );
  //priors
  //fixed effects
  mu 0 ~ normal(0,1);
  beta_1 ~ normal(0,1);
  //random effects
  P_cor ~ lkj_corr(2);
  to_vector(sd_P) ~ cauchy(0,1);
  sd_R \sim cauchy(0,1);
generated quantities {
  vector[2] V_P = sd_P .* sd_P;
  real V_R = sd_R * sd_R;
```

The model is now ready for estimation. We manually specify that the MCMC sampler should use 1000 iterations per chain to converge on the target joint posterior distribution warmup=1000, with the subsequent 1000 iterations used as posterior samples iter = 2000 (i.e. iter - warmup = number of MCMC samples per chain). init = 0 initializes the samplers near null values. Four

MCMC chains are used to assess model convergence across independent random samplers chains=4, with one core assigned to each chain for parallel processing cores=4. The adapt_delta=0.90 argument reduces the risk of divergent transitions during sampling.

Stan flags a few potential issues with the MCMC sampler. Further description of these and other warnings can be found in the Stan Warning Guide. One warning is that "The largest R-hat is NA, indicating chains have not mixed". Stan does not know whether some parameter values are fixed (causing Rhat = NA) because the sampler is stuck, or because the model has been intentionally specified with fixed parameter values (e.g. diagonals fixed to 1 in a correlation matrix or an intercept forced to 0). For the specified model, this is a harmless warning that can be safely ignored. However, we can also check for issues by looking at the Rhat values of all model parameters using summary() on the saved results. If an expected parameter is missing from the table or shows NA, this likely indicates an unintentional error in the model code.

```
summary(stan_results)$summary[,"Rhat"]
```

In addition to the Rhat warning, the effective sample sizes of some model parameters are too low to ensure accurate inferences. It is helpful to see which parameters are causing these warnings by sorting on the lowest n_eff values in the summary table.

```
sort(summary(stan_results)$summary[,"n_eff"])
```

It is typical that individual-specific trait values have relatively lower effective sample sizes than the population-level parameters of primary interest (e.g. the intercept deviation of individual 9 re_P[9,1] and their total intercept mu[9] compared to the global intercept mu_0). More damningly, however, we also see an extremely low effective sample for 1p__, which is the joint log density of the model (up to a constant internally defined scale factor. This provides further evidence that the model, as currently defined, is poorly identified. The key random effect SDs sd_P and variances V_P are also very poorly sampled, along with the residual SD sd_R and variance V_R. We could run the MCMC sampler for more iterations, increase the warm-up period, and change various other manual control settings. However, the deeper issue with the estimation procedure is not that the model is poorly defined or that the data provide insufficient information; rather, we have inefficiently parameterized the model.

1.4.1 Cholesky decompositions

Although the .stan file appropriately represents the formal model, it is programmed in such a way that the MCMC sampler has troubling sampling from the joint posterior distribution of the model. One of the first things we can do to increase efficiency is to reduce redundant computation over matrices in our model. This can be done with Cholesky decompositions. For any positive definite matrix Ω , a Cholesky decomposition can be defined such that

$$\Omega = \mathbf{L}_{\Omega} \mathbf{L}_{\Omega}^{\mathrm{T}}$$

where \mathbf{L}_{Ω} is a lower-triangular matrix and ^T indicates matrix transposition. This property means that we can always do computations of reduced dimensionality on the lower-triangular matrix \mathbf{L}_{Ω} and subsequently recover the full positive-definitive matrix Ω by post-multiplying \mathbf{L}_{Ω} with its transpose.

Stan provides many built-in functions for easily defining and manipulating Cholesky decomposed matrices, which we can use to re-parameterize the .stan file. Comments are added below where Cholesky decompositions have been introduced.

```
data {
  int<lower=0> N;
  int<lower=0> I;
  int<lower=0> id[N];
  vector[N] x;
  vector[N] y;
parameters {
  //fixed effects
  real mu_0;
  real beta_1;
  //random effects
  cholesky_factor_corr[2] LP_cor; //lower tri Cholesky of random effect cor matrix
  vector<lower=0>[2] sd_P;
  real<lower=0> sd_R;
  matrix[I,2] re_P;
}
transformed parameters {
  vector[I] mu = mu_0 + col(re_P, 1);
  vector[I] beta = beta_1 + col(re_P, 2);
  cholesky_factor_cov[2] LP = diag_pre_multiply(sd_P, LP_cor); //Cholesky of random effect cov
}
model {
```

```
//model likelihood
  y ~ normal(mu[id] + beta[id].*x, sd_R);
  for(i in 1:I)
 re_P[i] ~ multi_normal_cholesky([0,0], LP); //likelihood expecting Cholesky cov
  //priors
  //fixed effects
 mu 0 ~ normal(0,1);
 beta 1 ~ normal(0,1);
  //random effects
 LP_cor ~ lkj_corr_cholesky(2); //prior for Cholesky matrix
  to_vector(sd_P) ~ cauchy(0,1);
  sd_R \sim cauchy(0,1);
generated quantities {
  vector[2] V_P = sd_P .* sd_P;
  real V_R = sd_R * sd_R;
  corr_matrix[2] P_cor = LP_cor*LP_cor' ; //multiply by transpose to get full cor matr
  cov_matrix[2] P = diag_matrix(V_P) * P_cor * diag_matrix(V_P); //full cov matrix
```

The full covariance and correlation matrices are now specified in the generated quantities block.

1.4.2 Non-centered random effects

In addition to Cholesky decompositions, we can also re-parameterize the random effects to further enhance efficiency. Currently, we express the unobserved random effects in ${\tt re_P}$ as being generated from a distribution with unobserved (lower Cholesky) covariance matrix LP. While mathematically appropriate, this specification can make it difficult for the model to identify the scale of the random effects. An alternative but mathematically equivalent parameterization can be used to separating out the scale of the random effect deviations from the population-level (co)variances, which often will enhance model identification. Note that any normally distributed random variable z where

$$z \sim \text{Normal}(0, \sigma_z)$$

can also be expressed as a standard normal variable z_{std} scaled by the original SD

```
z \equiv z_{\mathbf{std}} \sigma_z
```

$$z_{\mathbf{std}} \sim \text{Normal}(0, 1)$$

Similarly for a $n\ge p$ matrix Z of p multivariate phenotypes with covariance matrix Σ

$$Z \equiv Z_{\mathbf{std}} \mathbf{L}_{\Sigma}$$

$$\text{vec}(Z_{\mathbf{std}}) \sim \text{MVNormal}(0, \mathbf{I})$$

where \mathbf{L}_{Σ} is the lower-triangular Cholesky decomposition. Implementing this so-called "non-centered parameterization" is straightforward in Stan. Building on the Cholesky decompositions added in the previous subsection

```
data {
  int<lower=0> N;
  int<lower=0> I;
  int<lower=0> id[N];
  vector[N] x;
  vector[N] y;
}
parameters {
  //fixed effects
 real mu_0;
  real beta_1;
  //random effects
  cholesky_factor_corr[2] LP_cor;
  vector<lower=0>[2] sd_P;
  real<lower=0> sd R;
  matrix[I,2] std_P; //now matrix of standard normals (see priors below)
}
transformed parameters {
  matrix[I,2] re_P = std_P * diag_pre_multiply(sd_P,LP_cor); //non-centered parameterization
  vector[I] mu = mu_0 + col(re_P, 1);
  vector[I] beta = beta_1 + col(re_P, 2);
}
model {
  //model likelihood
  y ~ normal(mu[id] + beta[id].*x, sd_R);
  //priors
  //fixed effects
  mu 0 ~ normal(0,1);
  beta_1 ~ normal(0,1);
```

```
//random effects
to_vector(std_P) ~ std_normal(); //new prior distribution over standard normal devia
LP_cor ~ lkj_corr_cholesky(2);
to_vector(sd_P) ~ cauchy(0,1);
sd_R ~ cauchy(0,1);

}
generated quantities {
  vector[2] V_P = sd_P .* sd_P;
  real V_R = sd_R * sd_R;
  corr_matrix[2] P_cor = LP_cor*LP_cor';
  cov_matrix[2] P = diag_matrix(V_P) * P_cor * diag_matrix(V_P);
}
```

Note that the specification of the random effects has been greatly simplified with the non-centered parameterization. By separating out the scale of the deviations and the population-level (co)variances, it becomes unnecessary to directly specify the generative distribution of the full random effects as above. Instead, the full distribution is partitioned into three independent priors over the random effect standard normal deviations, SDs, and correlations. This should make the model much easier to sample from.

The absence of warning messages indicates that our mathematically equivalent re-parameterizations have enhanced the efficiency of the MCMC sampler. The posterior samples of the model in MCMCsamples can subsequently be extracted, summarized, visualized, and manipulated. E.g.

```
post_beta_1 = MCMCsamples$beta_1 #extract population-level slope

median(post_beta_1) #central tendency of posterior
mad(post_beta_1) #dispersion around central tendency
quantile(post_beta_1, c(0.05,0.95)) #90% credible interval
```

```
sum(post_beta_1 > 0)/length(post_beta_1) #posterior probability of + effect
hist(post_beta_1) #MCMC approximation of posterior distribution
```

We encourage the use of the shinystan R package for deeper inspection of model convergence with a GUI prior to extraction of posterior results. In general, researchers should be skeptical of reporting results accompanied with sampler warnings and should seek to remove any diagnostic concerns prior to biological interpretation of the estimates.

1.5 Animal models

The model presented above assumes a single set of individual-specific intercepts and slopes, as defined by the mu and beta vectors in the .stan file. For quantitative genetic analysis with an animal model, these phenotypic effects can be further decomposed into distinct genetic and permanent environmental trait values. In particular, we expand the random phenotypic deviations so that

$$\begin{split} \mu_{\mathbf{P}} &= \mu_{\mathbf{A}} + \mu_{\mathbf{E}}, \quad \beta_{\mathbf{P}} = \beta_{\mathbf{A}} + \beta_{\mathbf{E}} \\ \begin{bmatrix} \mu_{\mathbf{A}} \\ \beta_{\mathbf{A}} \end{bmatrix} \sim \text{MVNormal}(0, \mathbf{G} \otimes \mathbf{A}) : \mathbf{G} &= \begin{bmatrix} \text{Var}(\mu_{\mathbf{A}}) & \text{Cov}(\mu_{\mathbf{A}}, \beta_{\mathbf{A}}) \\ \text{Cov}(\beta_{\mathbf{A}}, \mu_{\mathbf{A}}) & \text{Var}(\beta_{\mathbf{A}}) \end{bmatrix} \\ \begin{bmatrix} \mu_{\mathbf{E}} \\ \beta_{\mathbf{E}} \end{bmatrix} \sim \text{MVNormal}(0, \mathbf{E} \otimes \mathbf{I}) : \mathbf{E} &= \begin{bmatrix} \text{Var}(\mu_{\mathbf{E}}) & \text{Cov}(\mu_{\mathbf{E}}, \beta_{\mathbf{E}}) \\ \text{Cov}(\beta_{\mathbf{E}}, \mu_{\mathbf{E}}) & \text{Var}(\beta_{\mathbf{E}}) \end{bmatrix} \end{split}$$

where **A** is a positive-definite relatedness matrix derived from pedigree or molecular data. Two challenges arise when estimating such a model in Stan related to the computation of Kronecker products and the identification of genetic effects.

1.5.1 Kronecker products

There are no in-built Stan functions for computing Kronecker products \otimes . This could be overcome by manually specifying the Kronecker product function in the optional functions block of the model. Regardless, Kronecker products can be incredibly costly to compute, particularly for large matrices. It's thus desirable to find another alternative but mathematically equivalent parameterization to return random effects appropriately scaled by $\mathbf{G} \otimes \mathbf{A}$ without directly computing this term.

Fortunately, this can be easily accomplished by exploiting the properties of the matrix normal distribution, which generalizes the multivariate normal distribution to random variables described by matrices (Gupta and Nagar, 2018).

In particular, the matrix normal distribution for some $n \times p$ matrix **Y** of p phenotypes is given by

$$\mathbf{Y} \sim \mathrm{Matrix~Normal}_{n~x~p}(\mathbf{M}, \mathbf{U}, \mathbf{V})$$

where \mathbf{M} is a matrix of expected values and \mathbf{U} and \mathbf{V} are scaling matrices describing the among-row and among-column (co)variance respectively. This lesser known distribution generalizes from the multivariate normal distribution such that any matrix \mathbf{Y} will be matrix normally distributed if and only if

$$vec(\mathbf{Y}) \sim MVNormal_{np}(vec(\mathbf{M}), \mathbf{V} \otimes \mathbf{U})$$

where vec() is the vector operator, as used above in the model2.stan file. Given that we are interested in generating random effects with covariance $\mathbf{G}\otimes\mathbf{A}$, direct computation of the Kronecker product can be avoided by instead sampling the random effects from a matrix normal distribution with the appropriate scaling matrices, i.e. for the for the I x 2 matrix of additive genetic intercepts and slope deviations for I individuals

$$[\mu_{\mathbf{A}} \quad \beta_{\mathbf{A}}] \sim \text{Matrix Normal}_{\mathbf{I} \times 2}(\mathbf{0}, \mathbf{A}, \mathbf{G})$$

We can use the non-centered parameterization described above for the multivariate normal distribution to also more efficiently sample from this matrix normal distribution. In particular, a matrix $\mathbf{Z}_{\mathbf{I}_{\mathbf{x},\mathbf{z}}}$ can be defined for I individual standard normal deviations on each of 2 random effects, which are distributed such that

$$\mathbf{Z_{std}} \sim \mathrm{Matrix\ Normal_{I\ x\ 2}}(\mathbf{0}, \mathbf{I}, \mathbf{I})$$

The desired matrix of appropriately scaled, zero-centered random effects can then be defined such that

$$\begin{bmatrix} \mu_{\mathbf{A}} & \beta_{\mathbf{A}} \end{bmatrix} = 0 + \mathbf{L}_{\mathbf{A}} \mathbf{Z_{std}} \mathbf{L_{G}}^{\mathrm{T}}$$

where

$$\begin{bmatrix} \boldsymbol{\mu}_{\mathbf{A}} & \boldsymbol{\beta}_{\mathbf{A}} \end{bmatrix} \sim \text{Matrix Normal}_{\mathbf{I} \ \mathbf{x} \ 2}(0, \mathbf{L}_{\mathbf{A}} \mathbf{L}_{\mathbf{A}}^{\mathbf{T}}, \mathbf{L}_{\mathbf{G}} \mathbf{L}_{\mathbf{G}}^{\mathbf{T}})$$

As explained above, $\mathbf{L_A}$ is the lower triangular Cholesky decomposition of the \mathbf{A} matrix, while $\mathbf{L_G}^{\mathrm{T}}$ is the transpose of the lower triangular Cholesky decomposition of the \mathbf{G} covariance matrix. This sampling property of the matrix normal distribution therefore facilitates sampling from

$$\text{vec}(\begin{bmatrix} \mu_{\mathbf{A}} & \beta_{\mathbf{A}} \end{bmatrix}) \sim \text{MVNormal}_{\text{np}}(\text{vec}(\mathbf{0}), \mathbf{G} \otimes \mathbf{A})$$

}

through the multiplication of the $\mathbf{Z_{std}},\,\mathbf{L_A},\,$ and $\mathbf{L_G}^{\mathrm{T}}$ matrices.

This useful sampling property is straightforward to implement in Stan with appropriate data and can be used to account for any form of random effect covariation among individuals, which may extend beyond $\bf A$ alone. Thomson et al. (2018) provide an extensive review of various additional sources of autocorrelation that should be considered in quantitative genetic analyses.

This matrix normal approach is demonstrated in the SAM tutorial with simulated data. Here we review the relevant code in Stan to highlight how any relevant Kronecker product could be implemented. We build on the data simulation above by assuming that an additional relatedness matrix $\bf A$ is now available for partitioning the previously modelled phenotypic effects. The code of model2.stan can be modified accordingly, so that the mixed-effects model becomes an animal model for quantitative genetic analysis. The relatedness matrix $\bf A$ is declared in the data block, while the lower triangle Cholesky decomposition matrix $\bf L_A$ is generated in the transformed data block.

```
data {
  int<lower=0> N;
  int<lower=0> I;
  int<lower=0> id[N];
  vector[N] x;
  vector[N] y;
  matrix[I] A; //new relatedness matrix
transformed data{
  matrix[I,I] LA = cholesky decompose(A); //lower triangle relatedness matrix
New parameters are declared for the separate genetic (G) and permanent envi-
ronmental (E) effects.
parameters {
  //fixed effects
  real mu_0;
  real beta_1;
  //random effects
  cholesky_factor_corr[2] LG_cor; //additive genetic cor matrix
  cholesky_factor_corr[2] LE_cor; //permanent environmental cor matrix
  vector<lower=0>[2] sd_G; //SD of genetic effects
  vector<lower=0>[2] sd_E; //SD of environmental effects
  real<lower=0> sd_R;
  matrix[I,2] std_G; //matrix of standard normals for G effects
  matrix[I,2] std E; //matrix of standard normals for E effects
```

The appropriately scaled random deviations can then be specified in the transformed parameters block. The matrix normal parameterization is required for the additive genetic random effects, while the simpler non-centered approach may instead be used for the permanent environmental effects that are independently distributed among individuals. The ' function can also be used to return the transpose of the Cholesky decomposed covariance matrix $\mathbf{L_G} = \mathbf{G_{SD}} \mathbf{L_{GCar}}$ in Stan.

```
transformed parameters {
  matrix[I,2] re_G = LA * std_G * diag_pre_multiply(sd_G,LG_cor)' ; //matrix normal
  matrix[I,2] re_E = std_E * diag_pre_multiply(sd_E,LE_cor); //non-centered

vector[I] mu = mu_0 + col(re_G, 1) + col(re_E, 1); //P = G + E
  vector[I] beta = beta_1 + col(re_G, 2) + col(re_E, 2); //P = G + E
}
```

With the addition of new priors in the model block

```
to_vector(std_devG) ~ std_normal(); //standard normal deviates
to_vector(std_devE) ~ std_normal();
LG_cor ~ lkj_corr_cholesky(2);
LE_cor ~ lkj_corr_cholesky(2);
to_vector(sd_G) ~ cauchy(0,1);
to_vector(sd_E) ~ cauchy(0,1);
```

the model will be well defined and equivalent to the simpler formal model defined with Kronecker products of covariance matrices. Note that the permanent environmental effects are defined as they were for the purely phenotypic effects above, without consideration of the Kronecker product $\mathbf{E} \otimes \mathbf{I}$. This product indicates that individuals' trait values are independent and identically distributed, so that ignoring the Kronecker product in Stan with $\mathbf{re}_{\mathbf{E}} = \mathbf{std}_{\mathbf{E}} * \mathbf{diag}_{\mathbf{pre}_{\mathbf{multiply}}}(\mathbf{sd}_{\mathbf{E},\mathbf{LE}_{\mathbf{cor}}})$ is equivalent to specifying the matrix normal parameterization with additional Cholesky identity matrix $\mathbf{L}_{\mathbf{I}}$, i.e. $\mathbf{re}_{\mathbf{E}} = \mathbf{LI} * \mathbf{std}_{\mathbf{E}} * \mathbf{diag}_{\mathbf{pre}_{\mathbf{multiply}}}(\mathbf{sd}_{\mathbf{E},\mathbf{LE}_{\mathbf{cor}}})'$.

1.5.2 Identifying genetic effects

This matrix normal approach makes the animal model computationally efficient, but a more fundamental issue remains for identifying the scales of the distinct \mathbf{G} and \mathbf{E} effects during model estimation. Given that $\mathbf{P} = \mathbf{G} + \mathbf{E}$ under the assumption of independent additive effects, it can be difficult to uniquely identify the scale of the distinct genetic and environmental trait values, as any increase/decrease in genetic trait values can be compensated by an equivalent

decrease/increase in the environmental trait value to achieve equivalent phenotypic values. In principle, this issue is addressed by the fixed information in ${\bf A}$ that is provided to the model prior to estimation. In reality, however, relatedness matrices in the wild are often quite sparse, with most elements at or near 0. As a consequence, when a single individual-level parameter is expressed as the sum of two distinct parameters, as differentiated by the scaling of ${\bf A}$ and ${\bf I}$, it can be challenging to identify the proportion of variance attributable to each effect. Note that in the simplest case of completely unrelated individuals, i.e. ${\bf A} = {\bf I}$, genetic and environmental effects are completely confounded and cannot be uniquely identified without introducing further assumptions, as any combination of genetic and environmental values summing to the same value will fit the data equally well.

Fortunately, in spite of the empirical reality of sparse relatedness matrices, it is possible to parameterize an animal model in Stan so that even weakly identified genetic effects can be disentangled from environmental effects, using whatever information is provided by the fixed relatedness matrix and empirical data. This is accomplished by re-expressing the scale of the **G** and **E** effects not as independent parameters, but rather as dependent variances derived from their proportion of a common phenotypic variance parameter. In other words, the model only has to identify the scale of the total phenotypic trait values rather than attempting to identify two independent but potentially confounded random effect variances, i.e.

$$\mathrm{Var}(\mu_{\mathbf{P}}) = \frac{\mathrm{Var}(\mu_{\mathbf{A}})}{\mathrm{Var}(\mu_{\mathbf{P}})} \mathrm{Var}(\mu_{\mathbf{P}}) + \frac{\mathrm{Var}(\mu_{\mathbf{E}})}{\mathrm{Var}(\mu_{\mathbf{P}})} \mathrm{Var}(\mu_{\mathbf{P}})$$

$$\mathrm{Var}(\beta_{\mathbf{P}}) = \frac{\mathrm{Var}(\beta_{\mathbf{A}})}{\mathrm{Var}(\beta_{\mathbf{P}})} \mathrm{Var}(\beta_{\mathbf{P}}) + \frac{\mathrm{Var}(\beta_{\mathbf{E}})}{\mathrm{Var}(\beta_{\mathbf{P}})} \mathrm{Var}(\beta_{\mathbf{P}})$$

The additive genetic proportions can be conceptualized as reaction norm heritabilities for the intercept and slope parameters

$$h_{\mu}^2 = \frac{\mathrm{Var}(\mu_{\mathbf{A}})}{\mathrm{Var}(\mu_{\mathbf{B}})}$$

$$h_{\beta}^2 = \frac{\mathrm{Var}(\beta_{\mathbf{A}})}{\mathrm{Var}(\beta_{\mathbf{P}})}$$

Given that there are only two individual-level random effects, the proportion of variance attributable to environmental effects is necessarily $1-h_{\mu}^2$ and $1-h_{\beta}^2$ for intercepts and slopes respectively. This alternative parameterization is again mathematically equivalent to the previous model, but it is much easier for Stan to estimate appropriately.

To implement this trick, we respecify the model parameters, removing the distinct genetic and environmental SDs and replacing them with common phenotypic SD scale parameters and reaction norm heritability parameters, which can subsequently be used to scale the distinct genetic and environmental standard normal deviates and correlation matrices in the transformed parameters block. This final model is given by

```
data {
  int<lower=0> N;
  int<lower=0> I;
  int<lower=0> id[N];
  vector[N] x;
  vector[N] y;
  matrix[I] A; //relatedness matrix
transformed data{
  matrix[I,I] LA = cholesky_decompose(A); //lower tri Cholesky matrix
parameters {
  //fixed effects
  real mu 0;
  real beta_1;
  //random effects
  cholesky_factor_corr[2] LG_cor; //additive genetic cor matrix
  cholesky_factor_corr[2] LE_cor; //permanent environmental cor matrix
  vector<lower=0>[2] sd_P; //total phenotypic SD (removed distinct G and E SDs)
  real<lower=0> sd R;
  matrix[I,2] std_G; //matrix of standard normals for G effects
  matrix[I,2] std_E; //matrix of standard normals for E effects
  //RN heritability (proportion between 0 and 1)
  vector<lower=0,upper=1>[2] RN_h2;
transformed parameters {
  vector<lower=0>[2] sd_G; //SDs of G effects
  vector<lower=0>[2] sd_E; //SDs of E effects
  matrix[I,2] re_G; //scaled G random effects
  matrix[I,2] re_E; //scaled E random effects
  vector[I] mu; //total individual intercepts
  vector[I] beta; //total individual slopes
  //SDs of genetic effects, sqrt(phenotypic variance * h2)
  sd_G[1] = sd_P[1] * sqrt(RN_h2[1]); //genetic SD for ind intercepts
  sd_G[2] = sd_P[2] * sqrt(RN_h2[2]); //genetic SD for ind slopes
```

```
//SDs of environmental effects, sqrt(phenotypic variance * [1-h2])
  sd_E[1] = sd_P[1] * sqrt(1 - RN_h2[1]); //environment SD for ind intercepts
  sd_E[2] = sd_P[2] * sqrt(1 - RN_h2[2]); //environment SD for ind slopes
  //matrix normal parameterization
  re_G = LA * std_devG * diag_pre_multiply(sd_G, LG_cor)';
  //non-centered parameterization
  re_E = std_devE * diag_pre_multiply(sd_E, LE) ;
  //phenotypic RN effects, P = G + E
  re_P = re_G + re_E;
  //total trait values (+ population fixed effects)
  mu = mu_0 + col(re_G, 1) + col(re_E, 1); //P = G + E
  beta = beta_1 + col(re_G, 2) + col(re_E, 2); //P = G + E
}
model {
  //model likelihood
  y ~ normal(mu[id] + beta[id].*x, sd_R);
  //priors
  //fixed effects
  mu 0 ~ normal(0,1);
  beta_1 ~ normal(0,1);
  //random effects
  to_vector(std_G) ~ std_normal(); //genetic std normal deviates
  to_vector(std_E) ~ std_normal(); //environmental std normal deviates
  LG_cor ~ lkj_corr_cholesky(2); //genetic correlations
  LE_cor ~ lkj_corr_cholesky(2); //environmental correlations
  to_vector(sd_P) ~ cauchy(0,1); //only phenotypic scale
  sd_R \sim cauchy(0,1);
  //reaction norm heritability
  to_vector(RN_h2) ~ beta(1.2,1.2);
generated quantities {
matrix[2,2] Gcor = LG * LG'; //genetic cor
matrix[2,2] Ecor = LE * LG'; //environmental cor
matrix[2,2] Rcor = LR * LR'; //residual cor
matrix[2,2] Gcov = diag_matrix(sd_G)*Gcor*diag_matrix(sd_G); //genetic cov
matrix[2,2] Ecov = diag matrix(sd E)*Ecor*diag matrix(sd E); //environmental cov
matrix[2,2] Rcov = diag_matrix(sd_R)*Rcor*diag_matrix(sd_R); //residual cov
```

}


```
matrix[2,2] Pcov = Gcov + Ecov; //phenotypic covariance (assuming independent effects)
matrix[2,2] Pcor = inverse(diag_matrix(sd_P))*Pcov*inverse(diag_matrix(sd_P)); //phenor

//variances
vector<lower=0>[2] V_P = sd_P .* sd_P;
vector<lower=0>[2] V_G = sd_G .* sd_G;
vector<lower=0>[2] V_E = sd_E .* sd_E;
vector<lower=0>[2] V_R = sd_R .* sd_R;
```

Note that because we specify phenotypic SDs $\operatorname{sd_P}$, the genetic SDs $\operatorname{sd_G}$ are calculated as $\operatorname{sqrt}(\operatorname{Var}(\mu_{\mathbf{P}})h_{\mu}^2) = \operatorname{SD}(\mu_{\mathbf{P}})\operatorname{sqrt}(h_{\mu}^2)$ and $\operatorname{sqrt}(\operatorname{Var}(\beta_{\mathbf{P}})h_{\beta}^2) = \operatorname{SD}(\beta_{\mathbf{P}})\operatorname{sqrt}(h_{\beta}^2)$, with the same approach taken for the proportion of environmental effects $1-h_{\mu}^2$ and $1-h_{\beta}^2$. A weakly regularizing Beta(1.2,1.2) prior is placed on the reaction norm heritability parameters, which are constrained between 0 and 1. This and any other prior can be easily visualized in R by randomly sampling from the relevant distribution.

```
hist( rbeta(1e5, 1.2, 1.2), prob = TRUE )
```

Histogram of rbeta(1e+05, 1.2, 1.2)



This prior is therefore relatively flat and uninformative over the range of plausible values, but provides very weak regularization by giving lower relative probability at the extreme ends approaching 0 (no genetic effect) and 1 (complete

genetic effect). With more than two individual-level random effects, such as when specifying multiple matrices of individual autocorrelation (Thomson et al., 2018), SDs and variances can instead be parameterized as scaled simplexes.

Chapter 2

SAM coding tutorial

This section is a work in progress \dots

2.1 Coding the model

...

2.2 Quantifying assortment

...

2.3 Selection differentials

...

2.4 The response to selection

...

Chapter 3

Extending SAMs

This section is a work in progress. Future topics to cover include, among others, SAMs with non-Gaussian phenotypes and fitness measures, social effects due to groups larger than dyads, integrated trait- and variance-partitioning models, and hierarchical phenotypes and latent variable models.

Bibliography

- Butler, D. G., Cullis, B. R., Gilmour, A. R., Gogel, B. J., and Thompson, R. (2018). *ASReml-R Reference Manual*. VSN International Ltd, Hemel Hempstead, UK.
- Carpenter, B., Gelman, A., Hoffman, M. D., Lee, D., Goodrich, B., Betancourt, M., and Riddell, A. (2017). Stan: A probabilistic programming language. Journal of Statistical Software, 74.
- Gelman, A., Vehtari, A., Simpson, D., Margossian, C. C., Carpenter, B., Yao, Y., and Modrák, M. (2020). Bayesian workflow. *arXiv preprint*, arXiv:2011.01808.
- Gupta, A. K. and Nagar, D. K. (2018). Matrix variate distributions. CRC Press.
- Hadfield, J. D. (2010). Mcmc methods for multi-response generalized linear mixed models: the mcmcglmm r package. *Journal of Statistical Software*, 33.
- Lemoine, N. P. (2019). Moving beyond noninformative priors: why and how to choose weakly informative priors in bayesian analyses. *Oikos*, 128.
- Martin, J. S. and Jaeggi, A. V. (2021). Social animal models for quantifying plasticity, assortment, and selection on interacting phenotypes. *Journal of Evolutionary Biology*, XX.
- McElreath, R. (2020). Statistical rethinking: A Bayesian course with examples in R and Stan. CRC Press, 2 edition.
- Nishio, M. and Arakawa, A. (2019). Performance of hamiltonian monte carlo and no-u-turn sampler for estimating genetic parameters and breeding values. *Genetics Selection Evolution*, 51.
- R Core Team (2020). R: A Language and Environment for Statistical Computing. R Foundation for Statistical Computing, Vienna, Austria.
- Stinchcombe, J. R., Simonsen, A. K., and Blows, M. W. (2014). Estimating uncertainty in multivariate responses to selection. *Evolution*, 68.

36 BIBLIOGRAPHY

Thomson, C. E., Winney, I. S., Salles, O. C., and Pujol, B. (2018). A guide to using a multiple-matrix animal model to disentangle genetic and nongenetic causes of phenotypic variance. *PloS one*, 13.